
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc153535358]Corrections required
1.1 [bookmark: _Toc153535359]fl_TC_11_4_5_NR5GC_TestBody
	Function name
	fl_TC_11_4_5_NR5GC_TestBody

	Reason for change
	It is observed that a fast UE can transmit  De-registration REQ (Step 10Ea1) immediately after sending the PDU Session Release complete at Step 10A.

In current TTCN implementation, the De-registration REQ for such a fast device is received in default handler that gets triggered according to following flow:
f_TC_11_4_5_NR5GC  fl_TC_11_4_5_NR5GC_TestBody  f_NR5GC_EmergencyCallRelease  f_NR5GC_EmergencyCallReleaseMT  f_NR5GC_PDUSessionRelease  f_NR_RRCReconfigReleaseDRBs  f_NR_SS_CommonRadioBearerConfig  a_NR5GC_StandardDefault  a_NR_BASE_StandardDefault

In order to avoid this from happening, the TTCN implementation of Step 10A is re-worked. Please see below.

At Step 17a1-17b3a1, according to R5-237454 the Registration REQ can come integrity protected, or unprotected. The current TTCN implementation does not accommodate the case when it is integrity protected. This needs to be accommodated.
Please see below.

	Summary of change
	Reworked the TTCN implementation for Step 10A
Updated the TTCN at Step 17a1-17b3a1 to accommodate the case when Reigstration REQ is integrity protected.

	TTCN module
	Emergency_NR5GC.ttcn

	MCC160 Comment
	



Before Change:
	function fl_TC_11_4_5_NR5GC_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;
    var NR_SecurityParams_Type v_RRC_SecurityParams;
    var TypeOfRegistration_Type v_TypeOfRegistration;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var NG_NAS_UL_Pdu_Type v_RxdMsg;
    timer t_Wait := 5.0;
    //@siclog "Step 1" siclog@
    //The SS configures:NR Cell 11 as "Serving cell" NR Cell 1 as "Non-Suitable "Off" cell".//@sic R5s210875 R5-217820 sic@
    f_NR_SetCellPower (nr_Cell11, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2);
    f_NR_SetCellPower (nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE);
    //@siclog "Step 2" siclog@
    //The UE transmits an RRCSetupRequest message.    //@sic R5-216201 sic@
    //@siclog "Step 3" siclog@
    //SS transmit an RRCSetupRequest message.//@sic R5-216201 sic@
    //@siclog "Step 4" siclog@
    //The UE transmits an RRCSetupComplete message to confirm the successful completion of the connection establishment and a REGISTRATION REQUEST message indicating "mobility registration updating" is sent to update the registration of the actual tracking area.
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell11, tsc_NR_RRC_TI_Def, ?, tsc_SHT_IntegrityProtected);
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Req failed");
    }
    //@siclog "Step 5" siclog@
    //SS sends a REGISTRATION REJECT message containing 5GMM cause value = #15 (No suitable cells in tracking area).
    SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell11,
                                   tsc_NR_RbId_SRB1, // No SRB2 so must be on SRB1
                                   -,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_NG_REGISTRATION_REJECT(cs_GMM_GSM_Cause(omit, tsc_NR5GCCause_NoSuitableCellsInTA)))));
    //@siclog "Step 6" siclog@
    //The SS transmits an RRCRelease message.//@sic R5-216201 sic@
    f_NR_RRCRelease(nr_Cell11);
    //@siclog "Steps 7" siclog@
    //Make the UE attempt an IMS emergency call dialling an emergency number e.g. 112 or 911). (NOTE 1)
     f_UT_RequestIMSEmergencyCall(UT);//@sic R5s210875 sic@
    //@siclog "Step 8" siclog@
    //Check: Does the UE performs Generic Test Procedure for IMS Emergency call establishment without IMS Emergency registration as specified in TS 38.508-1 [4], subclause 4.9.11?
    f_NR5GC_EmergencyCallLimitedService(nr_Cell11);//@sic R5-216201 sic@
    f_NR_PreliminaryPass(__FILE__, __LINE__, "step 8");//@sic R5-216201 sic@
    //@siclog "Step 9-10" siclog@
    //@siclog "Step 10A" siclog@
    //@siclog "Step 10b-10E" siclog@// @sic R5-237454 sic@
    f_NR5GC_EmergencyCallRelease(nr_Cell11);// @sic R5-227575 R5-237454 sic@
    t_Wait.start;
    alt {
      [] SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell11, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
       {
          v_RxdMsg := v_ReceivedAsp.Signalling.Nas[0].Pdu;
          // The UE transmits a DEREGISTRATION REQUEST message with De-registration type IE set to "Normal de-registration".
          if (not f_Check_NG_DeregistrationReqMOMsg (v_RxdMsg.Msg, crs_DeregisterType('0'B, '0'B,'01'B))) {
             f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Deregistration Request failed");
          }
          //The SS replies with a DEREGISTRATION ACCEPT message.
          SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell11, tsc_NR_RbId_SRB2, -,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_NG_DEREGISTRATION_ACCEPT)));
        }
      [] t_Wait.timeout {}
      }
    //@siclog "Step 10B-12b1" siclog@
    //Void //@sic R5-225410 sic@
    //@siclog "Step 12A" siclog@
    //SS releases the RRC connection
    f_NR_RRCRelease(nr_Cell11);//@sic R5-216201 sic@//@sic R5s210875 sic@
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1],cms_IPCAN_IMS_Reset);//@sic R5s210875 R5s211354 sic@
    //@siclog "Step 12B" siclog@
    //Make the UE attempt an IMS non-emergency call.
    f_UT_RequestIMSCall(UT, -, LOCAL_CNF_REQUIRED);//@sic R5-216201 sic@//@sic R5s210875, R5s230789 sic@
    //@siclog "Step 12C" siclog@
    //Check: Does the UE transmit an RRCSetupRequest message for initial registration procedure on NR Cell 11 within the next 10 seconds?
    f_NR_CheckNoRRCSetupReq (nr_Cell11, 10.0);//@sic R5-216201 sic@//@sic R5s210875 sic@
    //@siclog "Step 13" siclog@
    //The SS configures: NR Cell 1 as "Suitable neighbour intra-frequency cell",NR Cell 11 as "Serving cell"//@sic R5-217820 sic@
    f_NR_SetCellPower (nr_Cell1, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR1, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR2);//@sic R5s210875 R5s211354 sic@
    f_NR_SetCellPower (nr_Cell11, tsc_NR_NonSuitableCellSSS_EPRE_FR1, tsc_NR_NonSuitableCellSSS_EPRE_FR2);//@sic R5-225410 R5-217820 sic@
    //@siclog "Step 14" siclog@
    //Check: Does the UE perform on NR Cell 1 the Registration procedure for initial registration as specified in TS 38.508-1 [4] subclause 4.5.2.2-2, 'connected without release'?
    SRB.check(receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest(cr_NR_EstablishmentCause_Any))));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 14 passed");

    f_NR5GC_MobileInfo_Init();//@sic R5s210875 sic@
    f_NR5GC_Security_Init();//@sic R5s210875 sic@
    v_RRC_SecurityParams := f_NR_Security_Get();//@sic R5s210875 sic@
    v_RRC_SecurityParams.AS_Integrity.Algorithm := px_NR_IntegrityProtAlgorithm;//@sic R5s210875 sic@
    v_RRC_SecurityParams.AS_Ciphering.Algorithm := px_NR_CipheringAlgorithm;//@sic R5s210875 sic@
    f_NR_Security_Set(v_RRC_SecurityParams);//@sic R5s210875 sic@
    v_TypeOfRegistration := f_NR5GC_InitialOrMobilityRegistrationRequest (v_GMM_MobilityInfo, v_ReceivedMsg, nr_Cell1);//@sic R5-225410 sic@
    //@siclog "17a1-17b3a1" siclog@  @sic R5-202600 sic@
    //NOTE: The 5GS registration type shall be only set as Mobility Registration for R16 UEs according to TS 24.501 subclause 5.2.3.2.5 specified in Release 16. The EXCEPTION description applies only to R15 UEs.
    if (v_TypeOfRegistration == Initial_NoSecurity) {
      //@siclog "17a1-17a16a1" siclog@
      //IF 5GS registration type is set as Initial Registration in step 22, THEN steps 5 to 20a1 of the registration procedure described in TS 38.508-1 [4] Table 4.5.2.2-2 are performed on NR Cell 12.
      f_NR5GC_RRC_Idle_Steps5_20(v_GMM_MobilityInfo, nr_Cell1, v_ReceivedMsg, TESTMode_OFF,-, v_TypeOfRegistration); //@sic R5-225410 sic@
    }
    else if (v_TypeOfRegistration == Mobility) {
      //@siclog "17b1-17b3a1" siclog@
      // IF 5GS registration type is set as Mobility Registration in step 22, THEN steps 5 to 6a1 of the generic test procedure in TS 38.508-1 [4] Table 4.9.5.2.2-1 are performed on NR Cell 11.
      f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell1, v_GMM_MobilityInfo); //@sic R5-225410 sic@
  }
  }



After Change:
	function fl_TC_11_4_5_NR5GC_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;
    var NR_SecurityParams_Type v_RRC_SecurityParams;
    var TypeOfRegistration_Type v_TypeOfRegistration;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var NG_NAS_UL_Pdu_Type v_RxdMsg;
    var template (value) NG_NAS_DL_Pdu_Type v_PiggyBackedMsgToSend; //WA#11_4_5
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo; //WA#11_4_5
    var template (omit) GSM_MobilityInfo_Type v_GSM_MobilityInfo_IMS; //WA#11_4_5
    var PDUSessionInfoList_Type v_PDUSessionInfo; //WA#11_4_5
    var O1_Type v_PDUSessionId; //WA#11_4_5
    var boolean v_FinishedIMS := false; //WA#11_4_5
    var PDN_Index_Type v_PDNIndex; //WA#11_4_5
    timer t_Wait := 5.0;
    //@siclog "Step 1" siclog@
    //The SS configures:NR Cell 11 as "Serving cell" NR Cell 1 as "Non-Suitable "Off" cell".//@sic R5s210875 R5-217820 sic@
    f_NR_SetCellPower (nr_Cell11, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2);
    f_NR_SetCellPower (nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE);
    //@siclog "Step 2" siclog@
    //The UE transmits an RRCSetupRequest message.    //@sic R5-216201 sic@
    //@siclog "Step 3" siclog@
    //SS transmit an RRCSetupRequest message.//@sic R5-216201 sic@
    //@siclog "Step 4" siclog@
    //The UE transmits an RRCSetupComplete message to confirm the successful completion of the connection establishment and a REGISTRATION REQUEST message indicating "mobility registration updating" is sent to update the registration of the actual tracking area.
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell11, tsc_NR_RRC_TI_Def, ?, tsc_SHT_IntegrityProtected);
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Req failed");
    }
    //@siclog "Step 5" siclog@
    //SS sends a REGISTRATION REJECT message containing 5GMM cause value = #15 (No suitable cells in tracking area).
    SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell11,
                                   tsc_NR_RbId_SRB1, // No SRB2 so must be on SRB1
                                   -,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_NG_REGISTRATION_REJECT(cs_GMM_GSM_Cause(omit, tsc_NR5GCCause_NoSuitableCellsInTA)))));
    //@siclog "Step 6" siclog@
    //The SS transmits an RRCRelease message.//@sic R5-216201 sic@
    f_NR_RRCRelease(nr_Cell11);
    //@siclog "Steps 7" siclog@
    //Make the UE attempt an IMS emergency call dialling an emergency number e.g. 112 or 911). (NOTE 1)
     f_UT_RequestIMSEmergencyCall(UT);//@sic R5s210875 sic@
    //@siclog "Step 8" siclog@
    //Check: Does the UE performs Generic Test Procedure for IMS Emergency call establishment without IMS Emergency registration as specified in TS 38.508-1 [4], subclause 4.9.11?
    f_NR5GC_EmergencyCallLimitedService(nr_Cell11);//@sic R5-216201 sic@
    f_NR_PreliminaryPass(__FILE__, __LINE__, "step 8");//@sic R5-216201 sic@
    //@siclog "Step 9-10" siclog@
    //@siclog "Step 10A" siclog@
    //@siclog "Step 10b-10E" siclog@// @sic R5-237454 sic@
    /*WA#11_4_5 COMMENTED OUT f_NR5GC_EmergencyCallRelease(nr_Cell11);// @sic R5-227575 R5-237454 sic@
    t_Wait.start;
    alt {
      [] SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell11, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
       {
          v_RxdMsg := v_ReceivedAsp.Signalling.Nas[0].Pdu;
          // The UE transmits a DEREGISTRATION REQUEST message with De-registration type IE set to "Normal de-registration".
          if (not f_Check_NG_DeregistrationReqMOMsg (v_RxdMsg.Msg, crs_DeregisterType('0'B, '0'B,'01'B))) {
             f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Deregistration Request failed");
          }
          //The SS replies with a DEREGISTRATION ACCEPT message.
          SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell11, tsc_NR_RbId_SRB2, -,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_NG_DEREGISTRATION_ACCEPT)));
        }
      [] t_Wait.timeout {}
      }*/
    //R&S Changes start    
    v_PDUSessionInfo := f_NR5GC_MobileInfo_GetSessionInfoList();
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS2], cms_IMS_IPCAN_Trigger("Trigger MO Call Release")); 
    //This is needed in case the UE didnt send a PDU Session Release REQ.
    v_GSM_MobilityInfo_IMS := f_Get_PDUSessionForDNNType(v_PDUSessionInfo, Emergency_PDN);
    v_GSM_MobilityInfo := valueof(v_GSM_MobilityInfo_IMS);
    v_PDUSessionId := v_GSM_MobilityInfo.SessionId;
    v_PDNIndex := v_GSM_MobilityInfo.PdnIndex;
    t_Wait.start;
    //Code to check if the UE sends a PDU Session Release REQ
    alt {
      [] IMS[tsc_Index_IMS2].receive(cmr_IMS_IPCAN_Trigger ("IMS Call Release Msgs Sent"))
       { 
         v_FinishedIMS := true;
         repeat;
       }
      //The UE transmits a PDU SESSION RELEASE REQUEST message.
      [] SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell11, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
       {
         t_Wait.stop; 
         if (not f_Check_NG_PDUSessionReleaseReq (v_GSM_MobilityInfo, v_PDUSessionInfo, v_ReceivedAsp.Signalling.Nas[0].Pdu, v_PDUSessionId))
          {
             f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Release Request failed");
          }         
         if (not (v_FinishedIMS)) {
           IMS[tsc_Index_IMS2].receive(cmr_IMS_IPCAN_Trigger ("IMS Call Release Msgs Sent"));
         }
       }      
      [] t_Wait.timeout
       {
          //Do nothing
       }
    }
    //Now send the PDU Session Release Command , this will happen even if UE did not send PDU Session Release REQ  
    v_PDUSessionId := v_GSM_MobilityInfo.SessionId;
    v_PiggyBackedMsgToSend := f_Get_NG_PDUSessionReleaseCommand(v_PDUSessionId,
                                                       cs_GMM_GSM_Cause (omit, '00011010'B)); 
    SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell11, tsc_NR_RbId_SRB2, -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_PiggyBackedMsgToSend)));
    
    //Check: Does the UE transmit a PDU SESSION RELEASE COMPLETE message?
    SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell11,
                                        tsc_NR_RbId_SRB2,
                                        cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered,tsc_SHT_IntegrityProtected)))) -> value v_ReceivedAsp;
    if (f_Check_NG_PDUSessionReleaseComplete (v_PDUSessionInfo, v_ReceivedAsp.Signalling.Nas[0].Pdu, v_PDUSessionId)) {
      f_NR_PreliminaryPass(__FILE__, __LINE__, "PDU Session Release complete received");
    } else {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Incorrect PDU Session Release complete");
    }
    //Now prepare to get the optional De-registration REQ
    t_Wait.start;
    alt {
      [] SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell11, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
       {
          v_RxdMsg := v_ReceivedAsp.Signalling.Nas[0].Pdu;
          t_Wait.stop;
          // The UE transmits a DEREGISTRATION REQUEST message with De-registration type IE set to "Normal de-registration".
          if (not f_Check_NG_DeregistrationReqMOMsg (v_RxdMsg.Msg, crs_DeregisterType('0'B, '0'B,'01'B))) {
             f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Deregistration Request failed");
          }
          //The SS replies with a DEREGISTRATION ACCEPT message.
          SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell11, tsc_NR_RbId_SRB2, -,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_NG_DEREGISTRATION_ACCEPT)));
        }
      [] t_Wait.timeout {}
    }
    f_IP_Handling_StopPDN(IP, cs_IP_StopPDN(valueof(v_PDNIndex)));
    f_IP_RtpRtcpLoopback(IP, v_PDNIndex, tsc_IMS_MediaPort_M1, omit);//@sic R5s230682 sic@
    f_NR5GC_MobileInfo_SetSessionInfoList (v_PDUSessionInfo);
    //R&S Changes stop
    //@siclog "Step 10B-12b1" siclog@
    //Void //@sic R5-225410 sic@
    //@siclog "Step 12A" siclog@
    //SS releases the RRC connection
    f_NR_RRCRelease(nr_Cell11);//@sic R5-216201 sic@//@sic R5s210875 sic@
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1],cms_IPCAN_IMS_Reset);//@sic R5s210875 R5s211354 sic@
    //@siclog "Step 12B" siclog@
    //Make the UE attempt an IMS non-emergency call.
    f_UT_RequestIMSCall(UT, -, LOCAL_CNF_REQUIRED);//@sic R5-216201 sic@//@sic R5s210875, R5s230789 sic@
    //@siclog "Step 12C" siclog@
    //Check: Does the UE transmit an RRCSetupRequest message for initial registration procedure on NR Cell 11 within the next 10 seconds?
    f_NR_CheckNoRRCSetupReq (nr_Cell11, 10.0);//@sic R5-216201 sic@//@sic R5s210875 sic@
    //@siclog "Step 13" siclog@
    //The SS configures: NR Cell 1 as "Suitable neighbour intra-frequency cell",NR Cell 11 as "Serving cell"//@sic R5-217820 sic@
    f_NR_SetCellPower (nr_Cell1, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR1, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR2);//@sic R5s210875 R5s211354 sic@
    f_NR_SetCellPower (nr_Cell11, tsc_NR_NonSuitableCellSSS_EPRE_FR1, tsc_NR_NonSuitableCellSSS_EPRE_FR2);//@sic R5-225410 R5-217820 sic@
    //@siclog "Step 14" siclog@
    //Check: Does the UE perform on NR Cell 1 the Registration procedure for initial registration as specified in TS 38.508-1 [4] subclause 4.5.2.2-2, 'connected without release'?
    SRB.check(receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest(cr_NR_EstablishmentCause_Any))));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 14 passed");

    f_NR5GC_MobileInfo_Init();//@sic R5s210875 sic@
    f_NR5GC_Security_Init();//@sic R5s210875 sic@
    v_RRC_SecurityParams := f_NR_Security_Get();//@sic R5s210875 sic@
    v_RRC_SecurityParams.AS_Integrity.Algorithm := px_NR_IntegrityProtAlgorithm;//@sic R5s210875 sic@
    v_RRC_SecurityParams.AS_Ciphering.Algorithm := px_NR_CipheringAlgorithm;//@sic R5s210875 sic@
    f_NR_Security_Set(v_RRC_SecurityParams);//@sic R5s210875 sic@
    v_TypeOfRegistration := f_NR5GC_InitialOrMobilityRegistrationRequest (v_GMM_MobilityInfo, v_ReceivedMsg, nr_Cell1);//@sic R5-225410 sic@
    //@siclog "17a1-17b3a1" siclog@  @sic R5-202600 sic@
    //NOTE: The 5GS registration type shall be only set as Mobility Registration for R16 UEs according to TS 24.501 subclause 5.2.3.2.5 specified in Release 16. The EXCEPTION description applies only to R15 UEs.
    if ((v_TypeOfRegistration == Initial_NoSecurity) or (v_TypeOfRegistration == Initial_Secure)) { //WA#11_4_5
      //@siclog "17a1-17a16a1" siclog@
      //IF 5GS registration type is set as Initial Registration in step 22, THEN steps 5 to 20a1 of the registration procedure described in TS 38.508-1 [4] Table 4.5.2.2-2 are performed on NR Cell 12.
      f_NR5GC_RRC_Idle_Steps5_20(v_GMM_MobilityInfo, nr_Cell1, v_ReceivedMsg, TESTMode_OFF,-, v_TypeOfRegistration); //@sic R5-225410 sic@
    }
    else if (v_TypeOfRegistration == Mobility) {
      //@siclog "17b1-17b3a1" siclog@
      // IF 5GS registration type is set as Mobility Registration in step 22, THEN steps 5 to 6a1 of the generic test procedure in TS 38.508-1 [4] Table 4.9.5.2.2-1 are performed on NR Cell 11.
      f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell1, v_GMM_MobilityInfo); //@sic R5-225410 sic@
  }
  }



